Всероссийская олимпиада по информатике 2010/2011 года (Новгородская область)

Решения первого тура муниципального этапа для 9-11 классов
Задача 1. Фокусник
Количество шаров в шляпе может изменяться следующим образом:

· если вытащено 2 белых шара, то 1 белый шар возвращается назад 
(W=W-1)

· если вытащено 2 черных шара, то в шляпу кладется один белый шар 
(B=B-2, W=W+1)
· если вытащен 1 белый и 1 черный шар, то черный шар возвращается в шляпу 
(W=W-1)
Следовательно, количество черных шаров может уменьшаться только на 2, поэтому если в шляпе нечетное количество черных шаров, то в шляпе останется один черный шар. Если количество черных шаров четно, то рано или поздно они все будут извлечены из шляпы и останутся только белые шары, что приведет к результату «последний шар белый».

Текст решения:

var w,b:longint;

begin

 assign(input, 'input.txt');

 reset(input);

 assign(output, 'output.txt');

 rewrite(output);

 readln(w, b);

 if (b mod 2 = 0) then

   writeln('WHITE')

 else

   writeln('BLACK');

end.

Задача 2. Сверхпростые числа
Ответом на задачу является число p[p[k]], где p – массив, содержащий простые числа. Таким образом, необходимо эффективно найти значения, которые нужно занести в этот массив. Количество простых чисел, необходимое для максимального значения входных данных равняется p[500] или 3571. Т.е. необходимо найти не более 3571 простого числа. При поиске простых чисел удобнее всего проверять очередного кандидата на простоту, проверяя делимость на уже найденные простые числа.

Текст решения:

const

 max = 5000;

var

 i,j,k:longint;

 p:array[1..max] of longint;

 n:longint;

 ok:boolean;

begin

 assign(input, 'input.txt');

 reset(input);

 assign(output, 'output.txt');

 rewrite(output);

 readln(k);

 p[1]:=2;

 n:=1;

 i:=3;

 while (n < max) do begin

  ok:=true;

  for j:=1 to n do

   if i mod p[j] = 0 then begin

    ok:=false;

    break;

   end;

  if ok then begin

   n:=n + 1;

   p[n]:=i;

  end;

  i:=i + 2;

 end;

 writeln(p[p[k]]);

end.

Задача 3. Две последовательности
Так как обе последовательности возрастающие, то очевидно, что все числа между ai и ai+1 попадают в последовательность bn. При этом разность между соседними членами последовательности an все время растет (из-за увеличения значений членов последовательности bn). Следовательно, если мы будем последовательно добавлять по одному члену к последовательности an, длина последовательности bn будет расти еще быстее. Поэтому можно поочередно находить значения ai, добавляя значения между ai-1 и ai в последовательность bn, при этом ai будет ссылаться на уже заведомо найденные bi-1 и bi-3.

Чтобы уложиться в ограничения по памяти, достаточно хранить только значения последовательности bn, а из последовательности an использовать только последние два значения, выполняя вычисления, пока i не достигнет n.
Текст решения:

const max=10000000;

var

  n, i, j, m, a, pa: longint;

  b: array[1..max] of longint;

begin

  assign(input, 'input.txt');

  reset(input);

  assign(output, 'output.txt');

  rewrite(output);

  readln(n);

  b[1]:=1;

  b[2]:=5;

  b[3]:=6;

  b[4]:=8;

  m:=3;

  pa:=7;

  for i:=5 to n do begin

   a:=b[i-1] + b[i-3];

   for j:=pa + 1 to a - 1 do begin

     if m >= n then break;

     m:=m + 1;

     b[m]:=j;

   end;

   pa:=a;

  end;

  if (n = 1) then a:=2;

  if (n = 2) then a:=3;

  if (n = 3) then a:=4;

  if (n = 4) then a:=7;

  writeln(a);

  writeln(b[n]);

end.

